
18 Solving Systems of Linear Differential Equations

This Chapter illustrates how Maple V’s built-in linear algebra package can be useful in solving linear systems with
constant coefficients.

We will use routines that come from the Maple V Library linalg. This library can be used in a session by invoking
the command:

> with(linalg):

Warning: new definition for norm

Warning: new definition for trace

Note: In the above statement we used : instead of ; to suppress output. The first example involves solving the
homogeneous linear system

X ′ = AX,

where A is the 3 by 3 matrix:

A :=
 −1 1 2

1 0 1
2 1 −1


The way to solve such a system involves the following steps:

1. Find the eigenvalues and eigenvectors.

2. Write the general solution.

The first step can sometimes be very laborious to solve by hand even for systems as small as 3 by 3. Maple
can be very useful in solving these problems. We now enter matrix.

> A := matrix(3,3,[-1,1,2,1,0,1,2,1,-1]);

A :=
 −1 1 2

1 0 1
2 1 −1


The command eigenvects gives the eigenvalues and corresponding eigenvectors in one step. If you enter the

following command then Maple will return the eigenvalues, that except for their order should be as follows:

> eigsA := eigenvects(A);

eigsA := [−1,1, { [ 1 − 2 1 ] } ], [−3,1, { [−1 0 1 ] } ], [ 2,1, { [ 1 1 1 ] } ]
The last expresson is a sequence of lists the first element of eigsA, eigsA[1], is the list

[−3,1, [−1,0,1]].

The first element of this list is given by eigsA[1][1] is -3 and is equal to one of the eigenvalues. The second ele-
ment of this list,eigsA[1][2], is the order of the eigenvalue (in this case 1). The final element in this list, eigsA[1][3],
is the set of eigenvectors corresponding to the eigenvalue, eigsA[1][1]. In this case there is only one eigenvector,

eigsA[1][3][1] = [−1,0,1].

Note: The order of the elements of eigsA vary from one time or another and thus sometime it might turn out
that eigsA[1] is

[−1,1, [1,−2,1]].

In another Maple seesion the last Maple V command might lead to a different order. Once we have eigenvalues
and enough independent eigenvectors to span the eigenspace of the A we can write the general solution.

> solA := sum(c[i]*exp(eigsA[i][1]*t)*eigsA[i][3][1],i=1..3);

sol A := c1 e(−t ) [ 1 − 2 1 ]+ c2 e(−3 t ) [−1 0 1 ]+ c3 e(2 t ) [ 1 1 1 ]

> solA := evalm(solA);

solA :=
[

c1 e(−t ) − c2 e(−3 t ) + c3 e(2 t ) − 2 c1 e(−t ) + c3 e(2 t ) c1 e(−t ) + c2 e(−3 t ) + c3 e(2 t )]
The Maple command, map, allows to verify that solA is indeed a solution. Substituting X = solA into X’ = A

X first involves differentiating solA.

> map(diff,solA,t);

[
− c1 e(−t ) + 3 c2 e(−3 t ) + 2 c3 e(2 t ) 2 c1 e(−t ) + 2 c3 e(2 t )

−c1 e(−t ) − 3 c2 e(−3 t ) + 2 c3 e(2 t )
]

The last result should equal A X.

> evalm(A &* solA);

[
− c1 e(−t ) + 3 c2 e(−3 t ) + 2 c3 e(2 t ) 2 c1 e(−t ) + 2 c3 e(2 t )

−c1 e(−t ) − 3 c2 e(−3 t ) + 2 c3 e(2 t )
]

Thus solA is a solution. Since all of the eigenvalues are distinct we have an eigenspace with the right dimension
and the solution is a general solution.

In applications we usually are required to solve initial value problems. The next thing we do is find the solution
of X’= A X that satisfies the initial conditions:

x1(0) = 0, x2(0) = −2, x3(0) = 1.

First we solve for the constants c1, c2, c3.

> sol := simplify(solve({subs(t=0,solA[1])=0,subs(t=0,solA[2])=-2,

> subs(t=0,solA[3])=1},{c[1],c[2],c[3]}));

sol :=
{

c2 = 1
2
, c3 = −1

3
, c1 = 5

6

}
We now substitute the values of c1, c2, and c3 into the general solution to obtain the desired particular solution:

> solAa := [subs(sol,solA[1]),subs(sol,solA[2]),subs(sol,solA[3])];

sol Aa :=
[

5
6

e(−t ) − 1
2

e(−3 t ) − 1
3

e(2 t ),− 5
3

e(−t ) − 1
3

e(2 t ),
5
6

e(−t ) + 1
2

e(−3 t ) − 1
3

e(2 t )

]
Once we know a particular solution we can ask for graphical information about the solution.For example a plot

of x1 versus time can be given by

> plot(solAa[1],t=-2..2,title = ‘x1 vs t‘);

-180

-160

-140

-120

-100

-80

-60

-40

-20

0
-2 -1 0 1 2

t

x1 vs t

We can use a parametric plot to plot x1 vs x2 or x1 vs x3 or any other pair of dependent variables.

> plot([solAa[1],solAa[2],t=-2..2],title = ‘x1 vs x2‘);

Since the solution curve lies in three dimensional space it is also useful to plot it using the Maple function
spacecurve which is contained in the package plots. Thus we invoke plots.

> with(plots):

> spacecurve([solAa[1],solAa[2],solAa[3]],t=-2..2,axes = BOXED);

Of course dsolve can be used to find both general and particular solutions:

> eq1 := D(x1)(t) =A[1,1]* x1(t)+A[1,2]*x2(t)+A[1,3]*x3(t);

eq1 := D( x1 )( t ) = −x1( t )+ x2( t )+ 2 x3( t )

> eq2 := D(x2)(t) =A[2,1]* x1(t) + A[2,2]*x2(t) +A[2,3]* x3(t);

eq2 := D( x2 )( t ) = x1( t )+ x3( t )

> eq3 := D(x3)(t) = A[3,1]*x1(t) + A[3,2]*x2(t) +A[3,3]* x3(t);

eq3 := D( x3 )( t ) = 2 x1( t )+ x2( t )− x3( t )

> dsolve({eq1,eq2,eq3},{x1(t),x2(t),x3(t)});

{x2( t ) = C1 e(2 t ) − 2 C3 e(−t ),x3( t ) = C1 e(2 t ) + C2 e(−3 t ) + C3 e(−t ),

x1( t ) = C1 e(2 t ) − C2 e(−3 t ) + C3 e(−t )}

> dsolve({eq1,eq2,eq3,x1(0)=0,x2(0)=-2,x3(0)=1},{x1(t),x2(t),x3(t)});

{
x2( t ) = − 5

3
e(−t ) − 1

3
e(2 t ),x1( t ) = 5

6
e(−t ) − 1

2
e(−3 t ) − 1

3
e(2 t ),

x3( t ) = 5
6

e(−t ) + 1
2

e(−3 t ) − 1
3

e(2 t )

}
The question that you should ask is: does this solution agree with the previous solution? Be careful, notice that

the last solution is a set of equations and their order is unimportant.

If the eigenvalues are complex then the procedure is a little more complicated. The next example shows how to
find the general solution of

X ′ = BX

where

B :=
 1 0 0

2 1 −2
3 2 1


First enter this matrix. Note: There are two different techniques for entering a 3 by 3 matrix into a Maple

session. Compare the command

> B := matrix([[1,0,0],[2,1,-2],[3,2,1]]);

we are about to use for entering B and the command

> A := matrix(3,3,[1,1,2,1,2,1,2,1,1]);

that we used for entering A. It seems that Maple treats both methods equally.

> B := matrix([[1,0,0],[2,1,-2],[3,2,1]]);

B :=
 1 0 0

2 1 −2
3 2 1


As before, the solution of such a system requires that we

1. Find the eigenvalues and eigenvectors.

2. Write the general solution.

Again we use the Maple command eigenvects.

> eigsB :=

> eigenvects(B);

eigsB :=
[

1,1,

{[
1
−3
2

1

]}]
,[

RootOf( Z2 − 2 Z + 5 ),1,

{[
0

1
2

RootOf( Z2 − 2 Z + 5 )− 1
2

1

]}]
The answer uses the notation

RootOf( Z2 − 2 Z + 5)

for the complex eigenvalues in this case. We will show how to deal with this in a moment, but first we will
write the solution corresponding to the real eigenvalue 1.

(Note: Remember that an item listed in a set is unordered. Hence the real eigenvalue may appear as eigsB[j][1]
for j= 1, 2, or 3. This particular time it occurs as the first element.)

> Sol[1]:= exp(eigsB[1][1]*t)*eigsB[1][3][1];

Sol1 := et

[
1
−3
2

1

]
> Sol[1] := evalm(");

Sol1 :=
[

et − 3
2

et et

]
Turning to the complex eigenvalues, we will use the command allvalues.

> complexeig := allvalues(eigsB[2][1]);

complexeig := 1+ 2 I,1− 2 I

Substitute one component of the result of allvalues for the RootOf expression.

> comvect := subs(eigsB[2][1]=complexeig[1],eigsB[2][3]);

comvect := { [ 0 I 1 ] }
> comsol := evalm(exp(complexeig[1]*t)*comvect[1]);

comsol := [ 0 I e( (1+2 I ) t ) e( (1+2 I ) t ) ]

Use Re and Im to obtain two real vector value solutions.

> Sol[2] := Re(comsol);

Sol2 := <( comsol )

> Sol[3] := Im(comsol);

Sol3 := =( comsol )

> evalm(Sol[2]);

[ 0 −=( e( (1+2 I ) t ) )<( e( (1+2 I ) t ) ) ]

Construct the general solution and use evalm and evalc to evaluate the answer.

> solB := evalm(sum(c[i]*Sol[i],i=1..3));

sol B :=
[

c1 et − 3
2

c1 et − c2=( e( (1+2 I ) t ) )+ c3<( e( (1+2 I ) t ) )

c1 et + c2<( e( (1+2 I ) t ) )+ c3=( e( (1+2 I ) t ) )

]
This does not look like a correct answer since Maple does not automatically simplify complex numbers. Thus

we must apply the Maple command evalc.

> solB := map(evalc,solB);

sol B :=
[

c1 et − 3
2

c1 et − c2 et sin(2 t )+ c3 et cos(2 t )

c1 et + c2 et cos(2 t )+ c3 et sin(2 t )

]
We can check to see if the equations are satisfied.

> map(diff,solB,t);

[
c1 et − 3

2
c1 et − c2 et sin(2 t )− 2 c2 et cos(2 t )+ c3 et cos(2 t )− 2 c3 et sin(2 t )

c1 et + c2 et cos(2 t )− 2 c2 et sin(2 t )+ c3 et sin(2 t )+ 2 c3 et cos(2 t )

]

> evalm(B &* solB);

[
c1 et − 3

2
c1 et − c2 et sin(2 t )− 2 c2 et cos(2 t )+ c3 et cos(2 t )− 2 c3 et sin(2 t )

c1 et + c2 et cos(2 t )− 2 c2 et sin(2 t )+ c3 et sin(2 t )+ 2 c3 et cos(2 t )

]
Let us now find a solution to an initial value problem and plot it.

> sol2 := simplify(solve({subs(t=0,solB[1])=
-1,subs(t=0,solB[2])=0,subs(t=0,solB[3])=1},{c[1],c[2],c[3]}));

sol2 :=
{

c1 = −1, c2 = 2, c3 = −3
2

}
> solBa := [subs(sol2,solB[1]),subs(sol2,solB[2]),subs(sol2,solB[3])];

sol Ba :=
[
−et,

3
2

et − 2 et sin(2 t )− 3
2

et cos(2 t ),−et + 2 et cos(2 t )− 3
2

et sin(2 t )

]
> plot(solBa[2],t=-Pi..Pi,title = ‘x2 vs t‘);

> plot([solBa[2],solBa[3],t=-Pi..Pi], title = ‘x2 vs x1‘);

> with(plots):

> spacecurve([solBa[1],solBa[2],solBa[3]],t=-Pi..Pi,axes = NORMAL);

As before we can solve the differential equation directly using dsolve. First we enter the equation in the proper
form.

> eqB1 :=diff(x1(t),t) = B[1,1]*x1(t)+B[1,2]*x2(t) +B[1,3]*x3(t);

eqB1 := ∂

∂t
x1( t ) = x1( t )

> eqB2 := diff(x2(t),t) = B[2,1]*x1(t)+B[2,2]*x2(t)+B[2,3]*x3(t);

eqB2 := ∂

∂t
x2( t ) = 2 x1( t )+ x2( t )− 2 x3( t )

> eqB3 := diff(x3(t),t) = B[3,1]*x1(t) + B[3,2]*x2(t)+B[3,3]*x3(t);

eqB3 := ∂

∂t
x3( t ) = 3 x1( t )+ 2 x2( t )+ x3( t )

> dsolve({eqB1,eqB2,eqB3},{x1(t),x2(t),x3(t)});

{
x3( t ) = C1 et + C2 et sin(2 t )+ C3 et cos(2 t ),

x2( t ) = − 3
2

C1 et − C3 et sin(2 t )+ C2 et cos(2 t ),x1( t ) = C1 et

}
> dsolve({eqB1,eqB2,eqB3,x1(0)=-1,x2(0)=0,x3(0)=1},{x1(t),x2(t),x3(t)});

{
x1( t ) = −et,x3( t ) = −et − 3

2
et sin(2 t )+ 2 et cos(2 t ),

x2( t ) = 3
2

et − 2 et sin(2 t )− 3
2

et cos(2 t )

}

Exercises 18.0

1. Enter the matrix

A :=
 2 1 1

2 2 −1
−8 −5 −2


into a Maple V worksheet.

(a) Use the Maple V command eigenvects to find the eigenvalues and corresponding eigenvectors of A.

(b) Write the general solution of the differential equation X ′ = AX.

2. Enter the matrix

B :=
 −3 2 3

0 −2 2
0 −2 −3


into a Maple V worksheet.

(a) Use the Maple command eigenvects,and allvalues to find the eigenvalues and corresponding real and
complex eigenvectors of B.

(b) Write the general solution of the differential equation X ′ = BX in real form.


